

CORSO DI LAUREA IN INGEGNERIA MECCANICA  
CORSO DI PROGETTAZIONE ASSISTITA DELLE STRUTTURE MECCANICHE

ESERCITAZIONE SULLA CONVERGENZA  
CONVERGENZA AUTOMATICA -EFFETTO DEL GRADO DELLA "SHAPE FUNCTION"  
MODELLO 2D

Data la lastra con intaglio mostrata nella Fig. 1, mettere a punto il relativo modello tramite tecniche di "Solid -Modelling", utilizzando l'approccio "BOTTOM-UP" e la tecnica "FREE-MESHING".

Con un modello di questo tipo è possibile effettuare uno studio di convergenza automatico attraverso il comando ADAPT. Il programma tenterà di ridurre al di sotto di un valore fissato il rapporto (espresso in percentuale) tra l'energia complessivamente associata all' errore in tutto il modello e quella totale.

Nel far questo, il programma utilizza come indicatore per la suddivisione degli elementi la "Standard Energy Norm" associata a ciascun elemento. In tal modo è possibile ottenere una convergenza di tipo "h" in modo automatico.

Una volta effettuato questo studio, verificare l'effetto della introduzione di elementi di ordine superiore, osservando come questi consentano l'ottenimento di precisioni analoghe a quelle dell'elemento 42, con modelli sensibilmente più grossolani.

Si tenti dapprima con l'elemento 82 che ha 8 nodi ed una funzione di forma quadratica, applicando le stesse tecniche di analisi utilizzate per l'elemento 42.

Si utilizzi infine l'elemento quadrilatero di tipo "p" STIFF145, capace di far uso di funzioni di forma fino all'ottavo grado. In questo caso non è più possibile usare il comando ADAPT, in quanto il programma effettua automaticamente uno studio di convergenza lasciando inalterato il "mesh" e variando invece il grado delle funzioni di forma dei diversi elementi ("p-convergence"). La convergenza può essere controllata attraverso il comando PCONV.

## ESEMPIO N. 1

```
FINISH
/CLEAR
/PREP7
C***** CORSO DI PROGETTAZIONE ASSISTITA *****
C***
C*** ESERCITAZIONE N° 2- MODELLO DI LASTRA CON INTAGLIO
C*** MODELLAZIONE SOLIDA PARAMETRICA
C***
C*** STUDIO DI CONVERGENZA AUTOMATICO -ELEMENTO 42 O 82
C***
C*** METODO "TOP-DOWN" -MESHATURA "FREE"
C***
C*** DEFINIZIONE PARAMETRI INIZIALI
C***
B=30
L=60
R=10
NA=8
NR=8
RAP1=2
NT=7
RAP2=3
C***
C*** INSERIMENTO PRIMITIVE
C***
RECTNG,0,B,0,L
PCIRC,0,R
C***
C*** OPERAZIONE DI SOTTRAZIONE DELLE PRIMITIVE
C***
BOPTN,KEEP,NO ! provoca la cancellazione delle entità di partenza dopo un'operazione Booleana
ASBA,1,2 ! sottrae l'area 2 dalla 1
C***
C*** DEF.NE TIPO DI ELEMENTO
C***
C*** (provare le due alternative)
C***
C*** ET,1,42
ET,1,82
MP,EX,1,210000
C***
C*** SUDDIVISIONE LINEE DI DEFAULT
SMRTSIZE,6
C***
C*** SUDDIVISIONE AREA
C***
AMESH,ALL
C***
C*** VINCOLI E CARICHI
C***
FINISH
/SOLU
LSEL,,LOC,X,B-0.1,B+0.1
DL,ALL,3, SYMM
LSEL,ALL
LSEL,,LOC,Y,-0.1,+0.1
DL,ALL,3, SYMM
LSEL,ALL
LSEL,,LOC,Y,L-0.1,L+0.1
```

```
SFL,ALL,PRES,-10
LSEL,ALL
C***
C*** SOLUZIONE
C***
C*** STUDIO DI CONVERGENZA "H" AUTOMATICA
C***
ADAPT,10,1
c*** Il primo parametro indica il numero di calcoli permesso, il secondo indica
C*** la percentuale massima ammessa di energia associata all'errore rispetto
C*** all'energia di deformazione totale del modello.
C*** (provare a variare quest'ultimo parametro)
C***
C*** POST-PROCESSING
C***
FINISH
/POST1
SET,1
PLNSOL,S,Y
```

## ESEMPIO 2

```
FINISH
/CLEAR
/PREP7
C***** CORSO DI PROGETTAZIONE ASSISTITA *****
C***
C*** ESERCITAZIONE N° 2- MODELLO DI LASTRA CON INTAGLIO
C*** MODELLAZIONE SOLIDA PARAMETRICA
C***
C*** STUDIO DI CONVERGENZA AUTOMATICA DI TIPO "P"
C***
C*** METODO "TOP-DOWN" -MESHATURA "FREE"
C***
C*** DEFINIZIONE PARAMETRI INIZIALI
C***
B=30
L=60
R=10
NA=8
NR=8
RAP1=2
NT=7
RAP2=3
C***
C*** INSERIMENTO PRIMITIVE
C***
RECTNG,0,B,0,L
PCIRC,0,R
C***
C*** OPERAZIONE DI SOTTRAZIONE DELLE PRIMITIVE
C***
BOPTN,KEEP,NO ! provoca la cancellazione delle entità di partenza dopo un'operazione Booleana
ASBA,1,2 ! sottrae l'area 2 dalla 1
C***
C*** DEF.NE TIPO DI ELEMENTO
C***
ET,1,145
MP,EX,1,210000
C***
C*** SUDDIVISIONE LINEE DI DEFAULT
C*** (verificare l'effetto dell'affinamento del "mesh")
SMRTSIZE,6
C***
C*** SUDDIVISIONE AREA
C***
AMESH,ALL
C***
C*** RICERCA IL NUMERO DEL NODO ALLA RADICE DELL'INTAGLIO TRAMITE
C*** IL COMANDO GET
C***
NSEL,,LOC,Y,-0.001,0.001
NSEL,R,LOC,X,R-0.001,R+0.001
*GET,NN,NODE,,NUM,MIN
NSEL,ALL
C***
C*** VINCOLI E CARICHI
C***
FINISH
/SOLU
LSEL,,LOC,X,B-0.1,B+0.1
DL,ALL,3, SYMM
LSEL,ALL
```

```
LSEL,,LOC,Y,-0.1,+0.1
DL,ALL,3, SYMM
LSEL,ALL
LSEL,,LOC,Y,L-0.1,L+0.1
SFL,ALL,PRES,-10
LSEL,ALL
C***
C*** SOLUZIONE
C***
C*** STUDIO DI CONVERGENZA "P" AUTOMATICA
C***
C*** CONTROLLO DELLA CONVERGENZA TRAMITE IL COMANDO PCONV
C*** Il primo parametro indica la variazione massima percentuale della grandezza
C*** di controllo tra due calcoli successivi, il secondo, terzo e quarto definiscono
C*** come grandezza di controllo la tensione sY nel nodo NN
C***
PCONV,0.01,S,Y,NN
C***
C*** POST-PROCESSING
C***
SOLVE
FINISH
/POST1
SET,1
PLNSOL,S,Y
*ASK,IFL,Premere return per continuare,0
PLCONV,ALL !rappresenta l'andamento del parametro di convergenza
*ASK,IFL,Premere return per continuare,0
PPLOT ! rappresenta il grado della f.ne di forma per i diversi elementi
```




Fig. 1 – Schema della lastra da analizzare




Fig. 2 – Risultato dello studio di “h-convergence” con elemento 42




Fig. 3 – Risultato dello studio di “h-convergence” con elemento 82




Fig. 3 – Risultato dello studio di “p-convergence” con elementi 145